
; This is a script inspired by Tag's Natural Wildlife script, for TES IV: Oblivion, April 2006. 
; The new script is designed to dynamically take into account disposition states, weapon drawn by PC, and new OOO Jephre's followers' faction.
; Also, the script now leaves the creature's aggression and confidence untouched, relying on other ways of controlling their behavior.
; Lastly, the script includes code to stop creatures from infinitely pursuing, based in their aggression and confidence values, a fleeing PC.
; Original: Tag! 05/2006 -- Present: Oscuro 08/2006

scn 3OCreatureDecisionScriptAI

; code lockout variables
short Decided
short Shy
short zone
short Cautious ; OOO's addition
short Aggressive ; OOO's addition
short Friendly ; OOO's addition
short IsInside ; OOO's addition

; Timer for checking the PC's distance to a shy, cautious or aggressive creature ; OOO's addition

float timer ; OOO's addition

; size of the "personal zone"
short zonesize

; Distance after which creature stops pursuing the player. ; OOO's addition
short ForgetDistance ; OOO's addition

; working variables
short abase
short cbase
short creset
short Pcharm ; OOO's addition
short PWill ; OOO's addition
short Pweap ; OOO's addition
short Pdisp ; OOO's addition
short Health ; OOO's addition
short DispReset ; OOO's addition
short AttackRoll ; OOO's addition
short FleeRoll1 ; OOO's addition
short FleeRoll2 ; OOO's addition
short KeenSense; OOO's addition

;**** Added a new OnLoad block to disable AI features when creatures are found in their dens ****

Begin OnLoad

AddSpell ZBuffSkillAcrobatUber

If IsInInterior == 1
Set IsInside to 1

else
Set IsInside to 0

endif

End 

Begin OnReset



AddSpell ZBuffSkillAcrobatUber

If IsInInterior == 1
Set IsInside to 1

else
Set IsInside to 0

endif

End 

;**** Changed the entire OnHit block. It now uses the new disposition variables and chances of fleeing ****

begin onHit Player
set Health to getBaseActorValue Health
set cbase to GetBaseActorValue Confidence
if Shy == 1

if Health <= ( GetActorValue Health / 2 )
if GetRandomPercent >= cbase

ModActorValue Aggression -100
ModActorValue Aggression abase
ModactorValue Confidence cbase
;Attacks!
StartCombat Player
Set Shy to 0
Set Aggressive to 1
Return

else
Return

endif
else

Return
endif

elseif Cautious == 1
if Health <= ( GetActorValue Health / 4 )

if ( GetRandomPercent >= ( cbase - 15 ) )
ModActorValue Aggression DispReset
ModActorValue Aggression 100
ModactorValue Confidence creset
Set Cautious to 0
Set Shy to 1
Return

else
Return

endif
else

Return
endif

elseif Aggressive == 1
if Health <= ( GetActorValue Health / 6 )

if GetRandomPercent >= cbase
ModActorValue Aggression DispReset
ModActorValue Aggression 100
ModactorValue Confidence creset
Set Aggressive to 0
Set Shy to 1
Return

else
Return

endif



else
Return

endif
endif
End

;**** Intoduced new timer block, modified heavily the "If zone 1/0" block, and completely reworked the "if decided 1/0" block ****

Begin GameMode

If GetDead == 1 || GetDistance Player >= 3000
if Decided != 1

Return
elseif Decided == 1

if shy == 1
if GetDistance Player >= ( ForgetDistance + 2048 ) || ( Player.HasMagicEffect CHML == 1 || Player.HasMagicEffect INVI == 1 || GetDetectionLevel Player == 0 || 

GetDetectionLevel Player == 1 )
ModActorValue Aggression -100
ModActorValue Aggression abase
ModActorValue Confidence cbase
set zone to 0
set decided to 0
set shy to 0
moddisposition Player DispReset
EvaluatePackage

endif
return

elseif cautious == 1
set zone to 0
set decided to 0
set cautious to 0
moddisposition Player DispReset
return

elseif aggressive == 1
StopCombat Player
set zone to 0
set decided to 0
set aggressive to 0
moddisposition Player DispReset
return

elseif friendly == 1
if GetIsCurrentPackage OOOZZZCreatureFollowPlayer

RemoveScriptPackage OOOZZZCreatureFollowPlayer
ModActorValue Aggression -10
ModActorValue Aggression abase
if OOOBeastLord > 0

Set OOOBeastLord to OOOBeastLord - 1
endif
set zone to 0
set decided to 0
set friendly to 0
moddisposition Player DispReset
return

endif
endif

endif
endif



If GetDead != 1
set abase to GetBaseActorValue Aggression
set cbase to GetBaseActorValue Confidence
set creset to -GetBaseActorValue Confidence
set DispReset to -GetBaseActorValue Aggression ; Addition to OOO's 1.3 Script
set Pcharm to Player.GetActorValue Personality ; Addition to OOO's 1.3 Script
set Pwill to Player.GetActorValue Willpower ; Addition to OOO's 1.3 Script
set Pweap to Player.IsWeaponOut ; Addition to OOO's 1.3 Script
set zonesize to (((abase + 40) * 4) - (cbase * 2)) + 200 ; Modified from 400 to 200 and cbase to ((abase + 20) * 4) - (cbase * 2), for OOO's 1.3 Script
set ForgetDistance to (( abase * 13 ) + ( cbase * 12 )) + ( 1000 ) ; Addition to OOO's 1.3 Script
set Pdisp to (( Pcharm / 4 ) + ( Pwill / 7 )) ; Addition to OOO's 1.3 Script

if decided == 1 ; Now OOO animals dynamically change their behavior towards players
Set Timer to Timer - GetSecondsPassed
if Timer <= -1

if shy == 1
if GetDistance Player >= ( ForgetDistance + 2048 ) || ( Player.HasMagicEffect CHML == 1 || Player.HasMagicEffect INVI == 1 || GetDetectionLevel Player == 0 || 

GetDetectionLevel Player == 1 )
ModActorValue Aggression -100
ModActorValue Aggression abase
ModActorValue Confidence cbase
set zone to 0
set decided to 0
set shy to 0
moddisposition Player DispReset
EvaluatePackage

endif
elseif cautious == 1

If GetDistance Player <= 900
if ( GetRandomPercent  >= ( cbase - 10 ) )

ModActorValue Aggression DispReset
ModActorValue Aggression 100
ModactorValue Confidence creset
;Flee
Set Cautious to 0
Set Aggressive to 0
Set Shy to 1

endif
elseif GetDistance Player >= ForgetDistance || ( Player.HasMagicEffect CHML == 1 || Player.HasMagicEffect INVI == 1 || GetDetectionLevel Player == 0 || 

GetDetectionLevel Player == 1 )
;message "cautious forgets you"
set zone to 0
set decided to 0
set cautious to 0
moddisposition Player DispReset

endif
elseif aggressive == 1

if IsInside != 1
if GetDistance Player >= ForgetDistance || ( Player.HasMagicEffect CHML == 1 || Player.HasMagicEffect INVI == 1 || GetDetectionLevel Player == 0 || 

GetDetectionLevel Player == 1 )
StopCombat Player
set zone to 0
set decided to 0
set aggressive to 0
moddisposition Player DispReset

endif
else

if ( Player.HasMagicEffect CHML == 1 || Player.HasMagicEffect INVI == 1 || GetDetectionLevel Player == 0 || GetDetectionLevel Player == 1 ) && ( GetDistance Player 
>= ( KeenSense * 10 ) )

StopCombat Player



set zone to 0
set decided to 0
set aggressive to 0
moddisposition Player DispReset

endif
endif

elseif friendly == 1
if GetDistance Player >= 3000 ; Get rid of OOO's friendly AI packages if distance to player prompts animal to leave

if GetIsCurrentPackage OOOZZZCreatureFollowPlayer
RemoveScriptPackage OOOZZZCreatureFollowPlayer
ModActorValue Aggression -10
ModActorValue Aggression abase
if OOOBeastLord > 0

Set OOOBeastLord to OOOBeastLord - 1
endif

endif
set zone to 0
set decided to 0
set friendly to 0
moddisposition Player DispReset

endif
endif

set Timer to 3
endif

endif
endif

; Attack if the PC gets too close, except if PC is using stealthy means. This section also uses OOO's new animal AI keen sense characteristic

if ( Player.HasMagicEffect CHML != 1 && Player.HasMagicEffect INVI != 1 ) || GetDetectionLevel Player == 3  || ( KeenSense >= 90 )
if (zone == 0)

if (decided == 1) && (GetDistance player <= zonesize) && ((cautious == 1) || (shy == 1)) && IsInCombat != 1
if Shy == 1

ModActorValue Aggression -100
ModActorValue Aggression abase
ModActorValue Confidence cbase
Set Shy to 0
Set Aggressive to 1
StartCombat Player

elseif Cautious == 1
if GetRandomPercent <= 50

Set Cautious to 0
Set Aggressive to 1
StartCombat Player

else
Set Cautious to 0
Set Shy to 1
ModActorValue Aggression DispReset
ModActorValue Aggression 100
ModactorValue Confidence creset

endif
endif
set zone to 1

endif
endif

endif



; Make a decision to become shy, aggressive, cautious or friendly

if GetDetected Player < 1
Return

elseif GetDetected Player >= 1 
if Decided == 0

if GetDistance Player < 2048

Set KeenSense to GetRandomPercent ; Addition to OOO's 1.3 Script
set AttackRoll to GetRandomPercent ; Addition to OOO's 1.3 Script
set FleeRoll1 to GetRandomPercent ; Addition to OOO's 1.3 Script
set FleeRoll2 to GetRandomPercent ; Addition to OOO's 1.3 Script

if IsInside == 1
Set Pdisp to Pdisp -20

endif

if OOOJephreFaction == 1 ; Having completed OOO's quests for Jephre boosts the likelihood of animals befriending players
set Pdisp to Pdisp + 40

endif

if Pweap == 1
set Pdisp to Pdisp - 10

endif

if abase >= Pdisp 
if ( AttackRoll <= abase )

StartCombat Player
Set Aggressive to 1
Set Shy to 0
Set Cautious to 0

else
ModDisposition Player abase
if ( FleeRoll1  >= ( cbase - 25 ) )

ModActorValue Aggression DispReset
ModActorValue Aggression 100
ModactorValue Confidence creset
;Flee
Set Shy to 1
Set Aggressive to 0
Set Cautious to 0

else
Set Cautious to 1
Set Aggressive to 0
Set Shy to 0

endif
endif

else ;if abase < Pdisp ; Here animals have a chance to get the proper values and AI packages so as to act friendly towards players
ModDisposition Player abase
if OOOJephreFaction == 1

Set Friendly to 1
Set Cautious to 0
Set Aggressive to 0
Set Shy to 0
if GetRandomPercent <= Pdisp

if OOOBeastLord < 3
ModActorValue Aggression DispReset
ModActorValue Aggression 10
AddScriptPackage OOOZZZCreatureFollowPlayer



Set OOOBeastLord to OOOBeastLord + 1
endif

endif
else ;if OOOJephreFaction == 0

if ( FleeRoll2  >= ( cbase - 20 ) )
ModActorValue Aggression DispReset
ModActorValue Aggression 100
ModactorValue Confidence creset
;Flee
Set Shy to 1
Set Cautious to 0
Set Aggressive to 0

else
Set Cautious to 1
Set Shy to 0
Set Aggressive to 0

endif
endif

endif
Set Decided to 1
endif

endif
endif

End


